
NONMEM code of the joint midazolam and 1’-hydroxymidazolam population pharmacokinetic model 

 
$INPUT    ID  ;individual number   

SEX   
AGE  
HT  ;height [m] 
WT  ;weight [kg]  
BMI  ;body mass index [kg/m2] 
TIME  
TINF  ;time of midazolam infusion [h] 
RATE  ;rate of midazolam infusion [ug/h] 
AMT  ;amount of midazolam administered [ug] 
ADM  ;order of administration 
EVID  
MDV  

     DV  
DVTYPE ;type of analyte  
CMT  ;compartment identifier    
FLAGMED ;flag according to perpetrator   

 
$SUBROUTINES ADVAN6 TOL=3 
 
$MODEL   NCOMP=2 
COMP=(CENTRAL,DEFOBS)   ;COMPARTMENT FOR MDZ 
COMP=(METAB)            ;COMPARTMENT FOR 1’-OH-MDZ 
 
$PK 
;------------------------------------------------------------------------- 
; Covariate impact: Time here is relative to midazolam administration 
;------------------------------------------------------------------------- 
;Placebo/no perpetrator 
IF(TIME.LT.2.OR.FLAGMED.EQ.1) CLFLAGMED = 1  ;placebo 
 
;[A] Efavirenz 
IF(TIME.GT.2.AND.TIME.LE.4.AND.FLAGMED.EQ.2) CLFLAGMED = ( 1 + THETA(6)) 
IF(TIME.GT.4.AND.TIME.LE.5.AND.FLAGMED.EQ.2) CLFLAGMED = ( 1 + THETA(7)) 
IF(TIME.GT.5.AND.TIME.LE.6.AND.FLAGMED.EQ.2) CLFLAGMED = ( 1 + THETA(8)) 
IF(TIME.GT.6.AND.TIME.LE.7.AND.FLAGMED.EQ.2) CLFLAGMED = ( 1 + THETA(9)) 
IF(TIME.GT.7.AND.TIME.LE.8.AND.FLAGMED.EQ.2) CLFLAGMED = ( 1 + THETA(10)) 
IF(TIME.GT.8.AND.FLAGMED.EQ.2) CLFLAGMED               = 1 
 
;[B] Rifampicin 
IF(TIME.LE.24.AND.FLAGMED.EQ.3) CLFLAGMED = 1 
IF(TIME.GT.24.AND.TIME.LE.26.AND.FLAGMED.EQ.3) CLFLAGMED = ( 1 + THETA(11)) 
IF(TIME.GT.26.AND.TIME.LE.28.AND.FLAGMED.EQ.3) CLFLAGMED = ( 1 + THETA(12)) 
IF(TIME.GT.28.AND.TIME.LE.30.AND.FLAGMED.EQ.3) CLFLAGMED = ( 1 + THETA(13)) 
IF(TIME.GT.30.AND.TIME.LE.32.AND.FLAGMED.EQ.3) CLFLAGMED = ( 1 + THETA(14)) 
IF(TIME.GT.32.AND.FLAGMED.EQ.3) CLFLAGMED                = ( 1 + THETA(15)) 
 
;[C] Voriconazole oral 
IF(TIME.GT.2.AND.TIME.LE.3.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(16)) 

IF(TIME.GT.3.AND.TIME.LE.4.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(17)) 
IF(TIME.GT.4.AND.TIME.LE.5.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(18)) 
IF(TIME.GT.5.AND.TIME.LE.6.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(19)) 
IF(TIME.GT.6.AND.TIME.LE.7.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(20)) 
IF(TIME.GT.7.AND.TIME.LE.8.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(21)) 
IF(TIME.GT.8.AND.TIME.LE.9.AND.FLAGMED.EQ.4) CLFLAGMED = ( 1 + THETA(22)) 
IF(TIME.GT.9.AND.FLAGMED.EQ.4) CLFLAGMED               = ( 1 + THETA(23)) 
 
;[D] Voriconazole i.v 
IF(TIME.GT.2.AND.TIME.LE.3.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(24)) 
IF(TIME.GT.3.AND.TIME.LE.5.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(25)) 
IF(TIME.GT.5.AND.TIME.LE.6.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(26)) 
IF(TIME.GT.6.AND.TIME.LE.7.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(27)) 
IF(TIME.GT.7.AND.TIME.LE.8.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(28)) 
IF(TIME.GT.8.AND.TIME.LE.9.AND.FLAGMED.EQ.5) CLFLAGMED = ( 1 + THETA(29)) 
IF(TIME.GT.9.AND.FLAGMED.EQ.5) CLFLAGMED               = ( 1 + THETA(30)) 
 
;------------- 
; Parent (MDZ) 
;------------- 
; Parent clearance 
TVCL = THETA(1) * CLFLAGMED 
CL   = TVCL * EXP(ETA(1)) 
; Parent volume of distribution 
TVV1 = THETA(2) 
V1   = TVV1 * EXP(ETA(2)) 
 
;---------------------- 
; Metabolite (1’-OH-MDZ) 
;---------------------- 
; Metabolite clearance 
TVCLM = THETA(3) 
CLM   = TVCLM * EXP(ETA(3)) 
; Metabolite volume of distribution 
TVV2  = THETA(4)          
V2    = TVV2 * EXP(ETA(4)) 
 
;Fraction metabolized 
FMET = THETA(5) 
 
S1   = V1/1000    ; change from mL to L 
S2   = V2/1000 
 
;-------------Rate constants----------------------------- 
k10 = (1-FMET) * CL/V1 
K12 = FMET * CL/V1        
K20 = CLM/V2 
 
 
$DES 
DADT(1) = -K10*A(1)-K12*A(1) 

DADT(2) =  K12*A(1)*(341.77/325.78)-K20*A(2)     ;(341.77/325.78)  
                                                 ; metabolite/parent ratio 
 
$ERROR 
IF(CMT.EQ.1) IPRED = A(1)/S1 
IF(CMT.EQ.2) IPRED = A(2)/S2 
 
;Proportional RUV 
W   = IPRED 
Y1  = IPRED + W*EPS(1)     ;parent 
Y2  = IPRED + W*EPS(2)     ;metabolite 
 
IF(CMT.EQ.1) TYPE=0   ;parent 
IF(CMT.EQ.2) TYPE=1   ;metabolite 
 
Y   = Y2*TYPE + Y1*(1-TYPE) 
 
IRES  = DV-IPRED 
DEL   = 0 
IF(IPRED.EQ.0)DEL = 0.0001 
IWRES = IRES/(W+DEL) 
 
;-------------Initial estimates---------------------------- 
$THETA (0,44)           ;1. CL_parent [L/h] 
$THETA (0.0001,56)      ;2. V_parent [L] 
$THETA (0,240)          ;3. CL_metabolite [L/h] 
$THETA (0.001,286)      ;4. V2_metabolite [L] 
$THETA  0.92 FIX        ;5. Fraction metabolised 
 
;A. cov efavirenz-CLFLAGMED2 
(0,0.15)  ; CLFLAGMED2  2-4 
(0,0.59)  ; CLFLAGMED2  4-5 
(0,0.54)  ; CLFLAGMED2  5-6 
(0,0.28)  ; CLFLAGMED2  6-7 
(0,0.33)  ; CLFLAGMED2  7-8 
 
;B. cov rifampicin-CLFLAGMED3 
(0,0.27) ; CLFLAGMED3  24-26 
(0,0.17) ; CLFLAGMED3  26-28 
(0,0.23) ; CLFLAGMED3  28-30 
(0,0.46) ; CLFLAGMED3  30-32 
(0,0.1)  ; CLFLAGMED3  32-36 
 
;C. cov voriconazole oral-CLFLAGMED4 
(-1,-0.4,0)   ; CLFLAGMED4  2-3 
(-1,-0.3,0)   ; CLFLAGMED4  3-4 
(-1,-0.23,0)  ; CLFLAGMED4  4-5 
(-1,-0.32,0)  ; CLFLAGMED4  5-6 
(-1,-0.64,0)  ; CLFLAGMED4  6-7 
(-1,-0.54,0)  ; CLFLAGMED4  7-8 
(-1,-0.70,0)  ; CLFLAGMED4  8-9 
(-1,-0.7,0)   ; CLFLAGMED4  9-10 

;D. cov voriconazole i.v-CLFLAGMED5 
(-1,-0.16,0)  ; CLFLAGMED5  2-3 
(-1,-0.11,0)  ; CLFLAGMED5  3-5 
(-1,-0.15,0)  ; CLFLAGMED5  5-6 
(-1,-0.36,0)  ; CLFLAGMED5  6-7 
(-1,-0.39,0)  ; CLFLAGMED5  7-8 
(-1,-0.6,0)   ; CLFLAGMED5  8-9 
(-1,-0.58,0)  ; CLFLAGMED5  9-10 
 
$OMEGA  0.04     ; IIV_CL_parent 
$OMEGA  0.09     ; IIV_V_parent 
$OMEGA  0.18     ; IIV_CL_metabolite 
$OMEGA  0.15     ; IIV_V2_metabolite 
 
$SIGMA  0.02     ; 1. Parent Prop RUV 
$SIGMA  0.05     ; 2. Metabolite Prop RUV 
 
$ESTIMATION METHOD=1 INTERACTION MAXEVAl=9999 SIG=3 NOABORT PRINT=5 
 
$COV 

 


